

Linear Mixed Effects Models Examples

Biostatistics 653

Applied Statistics III: Longitudinal Data Analysis

Evaluating the Need for Random Effects

- One way to think of the linear mixed effects model is to think of a random coefficient model in which every subject has its own intercept and slope. However, what if the slopes do not vary across units? In this case, we would think of slopes as fixed effects rather than random.

Evaluating the Need for Random Effects

Consider a simple example with only one group. The “full” model with random intercepts and slopes is given by

$$Y_{ij} = \beta_0 + \beta_1 t_{ij} + b_{0i} + b_{1i} t_{ij} + \epsilon_{ij}$$

where

$$V(\mathbf{b}_i) = \mathbf{D} = \begin{pmatrix} d_{11} & d_{12} \\ d_{12} & d_{22} \end{pmatrix}$$

If the slopes do not vary, then we have the “reduced” model given by

$$Y_{ij} = \beta_0 + \beta_1 t_{ij} + b_{0i} + \epsilon_{ij}$$

where

$$V(b_i) = d_{11}$$

Assume in each model $V(\epsilon_i) = \mathbf{R}_i = \sigma^2 \mathbf{I}_{n_i}$

Evaluating the Need for Random Effects

- Both models lead to the same specification for the mean $E(Y_i) = X_i\beta$, but they involve different covariance models $\Sigma_i = Z_i D Z_i^T + \sigma^2 I_{n_i}$.
- In the full model, Σ_i is constructed using

$$Z_i = \begin{pmatrix} 1 & t_{i1} \\ 1 & t_{i2} \\ \vdots & \vdots \\ 1 & t_{in_i} \end{pmatrix}$$

- In the reduced model, $D = d_{11}$, $Z_i = (1, \dots, 1)^T$ and Σ_i is compound symmetry

$$\Sigma_i = \begin{pmatrix} d_{11} + \sigma^2 & d_{11} & \cdots & d_{11} \\ d_{11} & d_{11} + \sigma^2 & \cdots & d_{11} \\ \vdots & \vdots & \ddots & \vdots \\ d_{11} & \cdots & d_{11} & d_{11} + \sigma^2 \end{pmatrix}$$

Evaluating the Need for Random Effects

- To determine which model is more appropriate, one might rely on information criteria (AIC, BIC) to select between these models. In addition, noticing that these are nested models, it would be natural to think about doing a likelihood ratio test!

Evaluating the Need for Random Effects

- We are often interested in comparing two nested models, one with q correlated random effects, and one with $q+1$ correlated random effects. The difference in covariance parameters between these two models is $q+1$, as we will need one additional variance and q additional covariances in the larger model. One might think about trying to conduct a likelihood ratio test then with $q + 1$ degrees of freedom, but there is a problem with this approach.

Evaluating the Need for Random Effects

- Testing for significance of random effects is not a standard hypothesis testing problem because the null $H_0: \mathbf{b}_i = 0$ is on the boundary (0) of the parameter space because variances cannot be negative. Thus the usual χ^2 asymptotic distribution of the LRT is invalid. However, it has been shown that to test whether a single random effect can be removed (e.g., using q random effects instead of $q + 1$), a mixture of two χ^2 distributions can be used to obtain p-values. In this case, one can calculate p-values based on $0.5 \chi_q^2 + 0.5 \chi_{q+1}^2$. Critical values are available in the Fitzmaurice book, Table C.1.

Evaluating the Need for Random Effects

- Thus if one wished to test whether a random intercepts model was sufficient for the dental data, compared to a model with random intercepts and a random slope for a linear term in age, one would conduct a LRT of the difference in the log-likelihoods and use a 50-50 mixture of a χ_1^2 and a χ_2^2 distribution to obtain the critical value.

Evaluating the Need for Random Effects

- In other situations, including more complex comparisons among nested models for covariance in which the null distribution of the LRT is not well understood (for example, comparing a model with q correlated random effects to one with $q + k$ correlated random effects, where $k > 1$), Fitzmaurice et al. recommend using $\alpha = 0.10$ with the df equal to the difference in number of covariance parameters (not the difference in number of random effects) instead of $\alpha = 0.05$ to prevent selection of a model that is too parsimonious; however you should be aware that they recommend this as an ad-hoc procedure (and that its theoretical properties have not been extensively studied).

Dental Example

- For the dental data, we can fit the random intercepts model and then use the mixture of chi-square distributions to determine if we need the random slope and random intercept, or if the random intercept only is sufficient.
- For the dental data, consider the reduced model

$$Y_{ij} = \beta_{0G} + \beta_{0B} + \beta_{1G}t_{ij} + \beta_{1B}t_{ij} + b_{0i} + \epsilon_{ij}$$

where

$$V(b_i) = d_{11}$$

and

$$V(\epsilon_i) = \sigma_B^2 \mathbf{I} \text{ or } \sigma_G^2 \mathbf{I}$$

SAS Code

```

title 'INTERCEPTS RANDOM FOR EACH GENDER';
title2 'R_i is diagonal within-child with gender-specific variance';
title3 'D is unstructured and the same for all children';
proc mixed method=ml data=proyuniv;
class newid gender;
model dist=gender gender*time/noint;
random intercept / type=un subject=newid g gcorr v vcorr;
repeated/group=gender subject=newid rcorr;
run;

```

| Model | Cov params | $-2 \log \hat{l}_{REML}$ | AIC | BIC |
|-------------------------------|------------|--------------------------|-------|-------|
| Random intercept and slope | 5 | 411.5 | 421.5 | 428.0 |
| Random intercept | 3 | 415.2 | 421.2 | 425.1 |

Dental Example

- In order to test whether we need the random slope for time, we compare the difference in -2 times the log-likelihoods to a mixture of a χ_1^2 and χ_2^2 distribution. With $\alpha = 0.05$, the critical value is 5.14. We calculate the value of the test statistic $415.2 - 411.5 = 3.7$, and we conclude that the random slope for time is not needed. The AIC and BIC criteria also prefer the model with only random intercepts. Although we used ML estimation here, the tests could also be carried out using REML.

Varying R and D for Dental Data

- Using the random intercepts and slopes model, we consider various D and Ri structures in the dental data. In particular, we have

$$Y_{ij} = \beta_0 + \beta_1 t_{ij} + b_{0i} + b_{1i} t_{ij} + \epsilon_{ij}$$

where $\mathbf{b}_i \sim N_2(0, \mathbf{D})$ and $\epsilon_i \sim N_4(0, \mathbf{R}_i)$.

- For each model, we will examine the number of parameters estimated, the log-likelihood, AIC, and BIC.

Varying R and D for Dental Data

Model 1 is the model fit previously in the slides, for which

$$V(\mathbf{b}_i) = \mathbf{D} = \begin{pmatrix} d_{11} & d_{12} \\ d_{12} & d_{22} \end{pmatrix}$$

and

$$V(\boldsymbol{\epsilon}_i) = \mathbf{R}_i = \begin{cases} \sigma_B^2 \mathbf{I}, & \text{for boys} \\ \sigma_G^2 \mathbf{I}, & \text{for girls} \end{cases}$$

We thus estimated 5 covariance parameters. The model had

$$\begin{aligned} -2\hat{l}_{REML} &= 411.5 \\ AIC &= 421.5 \\ BIC &= 428.0 \end{aligned}$$

Varying R and D for Dental Data

Model 2 has

$$V(\mathbf{b}_i) = \mathbf{D} = \begin{pmatrix} d_{11} & d_{12} \\ d_{12} & d_{22} \end{pmatrix}$$

and

$$V(\boldsymbol{\epsilon}_i) = \mathbf{R}_i = \sigma^2 \mathbf{I}$$

We thus estimated 4 covariance parameters. The model had

$$-2\hat{l}_{REML} = 432.6$$

$$AIC = 440.6$$

$$BIC = 445.8$$

SAS Code

```
title 'R_i diagonal, constant var sigma^2 for each gender';
title2 ' this is default R_i so no repeated statement needed ';
title3 'D is 2 x2 unstructured, same for both genders';
proc mixed data=proyuniv method=ml;
class newid gender;
model dist=gender gender*time/noint solution;
random intercept time/type=un subject=newid g gcorr v vcorr;
estimate 'diff in mean slope' gender 0 0 gender*time 1 -1;
contrast 'overall gender difference' gender 1 -1, gender*time 1 -1/e chisq;
run;
```


Varying R and D for Dental Data

Model 3 has

$$V(\mathbf{b}_i) = \mathbf{D} = \begin{pmatrix} d_{11} & d_{12} \\ d_{12} & d_{22} \end{pmatrix}$$

and

$$V(\boldsymbol{\epsilon}_i) = \mathbf{R}_i = \sigma^2 \begin{pmatrix} 1 & \rho & \rho^2 & \rho^3 \\ \rho & 1 & \rho & \rho^2 \\ \rho^2 & \rho & 1 & \rho \\ \rho^3 & \rho^2 & \rho & 1 \end{pmatrix}$$

We thus estimated 5 covariance parameters. The model had

$$-2\hat{l}_{REML} = 428.8$$

$$AIC = 438.8$$

$$BIC = 445.3$$

SAS Code

```
title 'R_i is AR(1), same var and rho for each gender';
title2 ' const var same for each gender ';
title3 'D is 2 x2 unstructured, same for both genders';
proc mixed data=proyuniv method=ml;
class newid gender;
model dist=gender gender*time/noint solution;
repeated / type=ar(1) subject=newid rcorr;
random intercept time/type=un subject=newid g gcorr v vcorr;
estimate 'diff in mean slope' gender 0 0 gender*time 1 -1;
contrast 'overall gender difference' gender 1 -1,
gender*time 1 -1/ e chisq;
run;
```

Varying R and D for Dental Data

Model 4 has

$$V(\mathbf{b}_i) = \mathbf{D}_h = \begin{pmatrix} d_{h11} & d_{h12} \\ d_{h12} & d_{h22} \end{pmatrix}$$

where $h=1,2$ indicates boys and girls, and

$$V(\boldsymbol{\epsilon}_i) = \mathbf{R}_i = \begin{cases} \sigma_B^2 \mathbf{I}, & \text{for boys} \\ \sigma_G^2 \mathbf{I}, & \text{for girls} \end{cases}$$

We thus estimated 8 covariance parameters. The model had

$$\begin{aligned} -2\hat{l}_{REML} &= 428.8 \\ AIC &= 438.8 \\ BIC &= 445.3 \end{aligned}$$

SAS Code

```
title 'R_i is diagonal within child but two vars';
title2 ' \sigma^2_G and \sigma^2_B';
title3 'D is 2 x2 unstructured, different for each gender';
proc mixed data=proyuniv method=ml;
class newid gender;
model dist=gender gender*time/noint solution;
repeated / group=gender subject=newid rcorr;
random intercept time/type=un group=gender
      subject=newid g gcorr v vcorr;
estimate 'diff in mean slope' gender 0 0 gender*time 1 -1;
contrast 'overall gender difference' gender 1 -1,
      gender*time 1 -1/
      e chisq;
run;
```

Varying R and D for Dental Data

Model 5 has

$$V(\mathbf{b}_i) = \mathbf{D} = \begin{pmatrix} d_{11} & d_{12} \\ d_{12} & d_{22} \end{pmatrix}$$

and

$$V(\boldsymbol{\epsilon}_i) = \mathbf{R}_i = \sigma_1^2 \mathbf{I} + \sigma^2 \begin{pmatrix} 1 & \rho & \rho^2 & \rho^3 \\ \rho & 1 & \rho & \rho^2 \\ \rho^2 & \rho & 1 & \rho \\ \rho^3 & \rho^2 & \rho & 1 \end{pmatrix}$$

We thus estimated 6 covariance parameters. The model had

$$\begin{aligned} -2\hat{l}_{REML} &= 428.7 \\ AIC &= 440.7 \\ BIC &= 448.5 \end{aligned}$$

SAS Code

```
title 'R_i sum of 2 components; an AR(1) component  
      for fluctuations ';  
title2 'and diagonal component with var sigma^2  
      common across gender';  
title3 'local option adds diag component to ar(1) structure';  
title4 'D is 2 x2 unstructured, samefor each gender';  
proc mixed data=proyuniv method=ml;  
class newid gender;  
model dist=gender gender*time/noint solution;  
repeated / type=ar(1) local subject=newid rcorr;  
random intercept time/type=un subject=newid g gcorr v vcorr;  
estimate 'diff in mean slope' gender 0 0 gender*time 1 -1;  
contrast 'overall gender difference' gender 1 -1,  
      gender*time 1 -1/ e chisq;  
run;
```

Degrees of Freedom for F Tests for Mixed Effects Models

- It is not always possible to find an exact F test for mixed and random effects models. When the sample size is reasonably large with respect to the number of parameters being estimated, we can rely on asymptotic results and base tests on the χ^2 distribution (which has an implicit assumption of infinite denominator degrees of freedom). In small samples, one can argue that the use of the χ^2 test is anti-conservative, as the asymptotic approximation can be bad with small N .

Degrees of Freedom for F Tests for Mixed Effects Models

- However, using the F can be questionable as well, as in general settings it may also involve approximations. Several approximations have been proposed, with those proposed by Satterthwaite (1946) and Kenward and Roger (1997) preferred. We note that these approximations are not the default in PROC MIXED, though the DDFM option may be used to ask for other approximations (DDFM=SATTERTH or DDFM=KENWARDROGER). With reasonably large N , we do not expect to see big differences in p-values across different approximations.

Degrees of Freedom for F Tests for Mixed Effects Models

- The Satterthwaite approximate F test in the mixed-factor ANOVA setting involves developing a linear combination of mean squares that has the same expectation as the mean squares of the factor effect of interest, when H_0 is true. This method is computationally intensive and may add substantially to computing time.

Degrees of Freedom for F Tests for Mixed Effects Models

- The method developed by Kenward and Roger is based on creating an estimated variance-covariance matrix that better approximates the truth in the small sample setting. In particular, they estimate the amount by which the asymptotic variance-covariance matrix underestimates (in a matrix sense) $V(\hat{\mathbf{b}})$. They then inflate the variance-covariance matrix by this amount and compute the degrees of freedom based on this adjustment, using a Satterthwaite type approach. This method has been shown to have nice properties in a number of simulation studies.

Degrees of Freedom for F Tests for Mixed Effects Models

- The default method for calculating degrees of freedom in PROC MIXED depends on the setting but generally takes less time to compute than the above methods. (The defaults are not problematic in certain simple settings, but for complex covariance structures with relatively small N one may wish to consider using the Kenward and Roger method instead.)

BLUPs for the Dental Data

Using the random intercepts and slopes model with \mathbf{D} unstructured and separate $\mathbf{R}_i = \sigma_h^2 \mathbf{I}_{n_i}$ for each gender, we now obtain the estimated BLUPs.

```
/* outpred gets BLUPs for individual estimated means */
/* ODS statement in second call produces datasets containing fixed */
/* effect estimates and the BLUPs for individual slopes and intercepts */

title 'R_i separate for each gender and diagonal ';
title2 'D unstructured';
proc mixed data=proyuni method=ml;
class newid gender;

model dist=gender gender*time/noint solution outpred=pdata;
repeated/group=gender subject=newid;
random intercept time/type=un subject=newid solution;
run;
proc print data=pdata; run;
```

BLUPs for the Dental Data

```
title 'R_i separate for each gender and diagonal ';
title2 'D unstructured';
proc mixed data=proyuniw method=ml;
class newid gender;
model dist=gender time*gender/noint solution;
repeated/group=gender subject=newid;
random intercept time/type=un subject=newid solution;
ods listing exclude SolutionF;
ods output SolutionF=fixedi;
ods listing exclude SolutionR;
ods output SolutionR=randi;
run;

data fixedi; set fixedi;
keep gender effect estimate;
run;

title3 'FIXED EFFECTS data';
proc print data=fixedi; run;

proc sort data=fixedi; by gender; run;
```

BLUPs for the Dental Data

```
data fixed12; set fixed1; by gender;
retain fixint fixslope;
if effect='GENDER' then fixint=estimate;
if effect='time*GENDER' then fixslope=estimate;
if last.GENDER then do;
output; fixint=.; fixslope=.;
end;
drop effect estimate;
run;

title3 'RECONFIGURED FIXED EFFECTS DATA';
proc print data=fixed12; run;

data rand1; set rand1;
gender=1; if newid<12 then gender=0;
keep newid gender effect estimate;
run;

title3 'random effects output data';
proc print data=rand1; run;
```

BLUPs for the Dental Data

```
proc sort data=rand1; by newid; run;

data rand12; set rand1; by newid;
retain ranint ranslope;
if effect='Intercept' then ranint=estimate;
if effect='time' then ranslope=estimate;
if last.newid then do;
output; ranint=.; ranslope=.;
end;
drop effect estimate;
run;

proc sort data=rand12; by gender newid; run;

title3 'reconfigured RE dataset';
proc print data=rand12; run;

data both1; merge fixed12 rand12; by gender;
beta0i=fixint+ranint;
betaii=fixslope+ranslope;
run;

title3 'random intercepts and slopes';
proc print data=both1; run;
```

BLUPs for the Dental Data

```

Model Information

Data Set                WORK.PROYUNIV
Dependent Variable      dist
Covariance Structures   Unstructured, Variance
                        Components
Subject Effects         newid, newid
Group Effect            GENDER
Estimation Method       ML
Residual Variance Method None
Fixed Effects SE Method Model-Based
Degrees of Freedom Method Containment
  
```

```

Class Level Information

Class    Levels    Values
newid      27      1 2 3 4 5 6 7 8 9 10 11 12 13
                        14 15 16 17 18 19 20 21 22 23
                        24 25 26 27
GENDER      2      0 1
  
```

```

Dimensions
Covariance Parameters      5
Columns in X                4
Columns in Z Per Subject    2
Subjects                    27
Max Obs Per Subject         4
  
```

```

Number of Observations

Number of Observations Read      108
Number of Observations Used      108
Number of Observations Not Used    0
  
```


BLUPs for the Dental Data

| Iteration History | | | |
|-------------------|-------------|--------------|------------|
| Iteration | Evaluations | -2 Log Like | Criterion |
| 0 | 1 | 478.24175986 | |
| 1 | 2 | 418.92503842 | 1.16632499 |
| 2 | 1 | 416.18869903 | 1.23326209 |
| 3 | 1 | 407.89638533 | 0.01954268 |
| 4 | 2 | 406.88264563 | 0.00645800 |
| 5 | 1 | 406.10632159 | 0.00056866 |
| 6 | 1 | 406.04318997 | 0.00000764 |
| 7 | 1 | 406.04238894 | 0.00000000 |

| Convergence criteria met. | | | |
|--------------------------------|---------|----------|----------|
| Covariance Parameter Estimates | | | |
| Cov Parm | Subject | Group | Estimate |
| UN(1,1) | newid | | 3.1978 |
| UN(2,1) | newid | | -0.1103 |
| UN(2,2) | newid | | 0.01976 |
| Residual | newid | GENDER 0 | 0.4449 |
| Residual | newid | GENDER 1 | 2.6294 |

| Fit Statistics | |
|--------------------------|-------|
| -2 Log Likelihood | 406.0 |
| AIC (smaller is better) | 424.0 |
| AICC (smaller is better) | 425.9 |
| BIC (smaller is better) | 435.7 |

BLUPs for the Dental Data

Null Model Likelihood Ratio Test

| DF | Chi-Square | Pr > ChiSq |
|----|------------|------------|
| 4 | 72.20 | <.0001 |

Solution for Fixed Effects

| Effect | GENDER | Estimate | Standard Error | DF | t Value | Pr > t |
|-------------|--------|----------|----------------|----|---------|---------|
| GENDER | 0 | 17.3727 | 0.7386 | 54 | 23.52 | <.0001 |
| GENDER | 1 | 16.3406 | 1.1114 | 54 | 14.70 | <.0001 |
| time*GENDER | 0 | 0.4795 | 0.06180 | 54 | 7.76 | <.0001 |
| time*GENDER | 1 | 0.7844 | 0.09722 | 54 | 8.07 | <.0001 |

Solution for Random Effects

| Effect | newid | Estimate | Std Err | DF | t Value | Pr > t |
|-----------|-------|----------|---------|----|---------|---------|
| Intercept | 1 | -0.4853 | 1.1744 | 54 | -0.41 | 0.6811 |
| time | 1 | -0.06820 | 0.1017 | 54 | -0.67 | 0.5052 |
| Intercept | 2 | -1.1922 | 1.1744 | 54 | -1.02 | 0.3146 |
| time | 2 | 0.1420 | 0.1017 | 54 | 1.40 | 0.1683 |
| Intercept | 3 | -0.8535 | 1.1744 | 54 | -0.73 | 0.4705 |
| time | 3 | 0.1773 | 0.1017 | 54 | 1.74 | 0.0869 |
| Intercept | 4 | 1.7024 | 1.1744 | 54 | 1.45 | 0.1530 |
| time | 4 | 0.04017 | 0.1017 | 54 | 0.40 | 0.6943 |
| Intercept | 5 | 0.9136 | 1.1744 | 54 | 0.78 | 0.4400 |
| time | 5 | -0.08680 | 0.1017 | 54 | -0.85 | 0.3970 |
| Intercept | 6 | -0.6740 | 1.1744 | 54 | -0.57 | 0.5684 |
| time | 6 | -0.07292 | 0.1017 | 54 | -0.72 | 0.4763 |
| Intercept | 7 | -0.05461 | 1.1744 | 54 | -0.05 | 0.9631 |
| time | 7 | 0.03641 | 0.1017 | 54 | 0.36 | 0.7217 |

BLUPs for the Dental Data

| | | | | | | |
|-----------|----|----------|--------|----|-------|--------|
| Intercept | 8 | 1.9350 | 1.1744 | 54 | 1.65 | 0.1052 |
| time | 8 | -0.1149 | 0.1017 | 54 | -1.13 | 0.2636 |
| Intercept | 9 | -0.2190 | 1.1744 | 54 | -0.19 | 0.8528 |
| time | 9 | -0.1151 | 0.1017 | 54 | -1.13 | 0.2624 |
| Intercept | 10 | -2.9974 | 1.1744 | 54 | -2.55 | 0.0136 |
| time | 10 | -0.09085 | 0.1017 | 54 | -0.89 | 0.3755 |
| Intercept | 11 | 1.9249 | 1.1744 | 54 | 1.64 | 0.1070 |
| time | 11 | 0.1530 | 0.1017 | 54 | 1.50 | 0.1382 |
| Intercept | 12 | 1.3469 | 1.4342 | 54 | 0.94 | 0.3519 |
| time | 12 | 0.08788 | 0.1232 | 54 | 0.71 | 0.4786 |
| Intercept | 13 | -0.8676 | 1.4342 | 54 | -0.60 | 0.5478 |
| time | 13 | -0.04068 | 0.1232 | 54 | -0.33 | 0.7424 |
| Intercept | 14 | -0.3575 | 1.4342 | 54 | -0.25 | 0.8041 |
| time | 14 | -0.02176 | 0.1232 | 54 | -0.18 | 0.8605 |
| Intercept | 15 | 1.5946 | 1.4342 | 54 | 1.11 | 0.2711 |
| time | 15 | -0.02772 | 0.1232 | 54 | -0.23 | 0.8228 |
| Intercept | 16 | -1.1581 | 1.4342 | 54 | -0.81 | 0.4229 |
| time | 16 | -0.04153 | 0.1232 | 54 | -0.34 | 0.7373 |
| Intercept | 17 | 0.8972 | 1.4342 | 54 | 0.63 | 0.5342 |
| time | 17 | 0.02260 | 0.1232 | 54 | 0.18 | 0.8551 |
| Intercept | 18 | -0.6889 | 1.4342 | 54 | -0.48 | 0.6329 |
| time | 18 | -0.02853 | 0.1232 | 54 | -0.23 | 0.8177 |
| Intercept | 19 | -0.1443 | 1.4342 | 54 | -0.10 | 0.9202 |
| time | 19 | -0.07348 | 0.1232 | 54 | -0.60 | 0.5533 |
| Intercept | 20 | -0.1273 | 1.4342 | 54 | -0.09 | 0.9296 |
| time | 20 | 0.02544 | 0.1232 | 54 | 0.21 | 0.8372 |
| | | | | | | |
| Intercept | 21 | 2.5349 | 1.4342 | 54 | 1.77 | 0.0828 |
| time | 21 | 0.1088 | 0.1232 | 54 | 0.88 | 0.3811 |
| Intercept | 22 | -0.2261 | 1.4342 | 54 | -0.16 | 0.8753 |
| time | 22 | -0.08535 | 0.1232 | 54 | -0.69 | 0.4913 |
| Intercept | 23 | -0.6374 | 1.4342 | 54 | -0.44 | 0.6585 |
| time | 23 | 0.006510 | 0.1232 | 54 | 0.05 | 0.9580 |
| Intercept | 24 | -1.7008 | 1.4342 | 54 | -1.19 | 0.2409 |
| time | 24 | 0.1139 | 0.1232 | 54 | 0.92 | 0.3591 |
| Intercept | 25 | 0.2387 | 1.4342 | 54 | 0.17 | 0.8684 |
| time | 25 | -0.03166 | 0.1232 | 54 | -0.26 | 0.7981 |
| Intercept | 26 | 0.1180 | 1.4342 | 54 | 0.08 | 0.9347 |
| time | 26 | 0.06104 | 0.1232 | 54 | 0.50 | 0.6222 |
| Intercept | 27 | -0.8228 | 1.4342 | 54 | -0.57 | 0.5688 |
| time | 27 | -0.07545 | 0.1232 | 54 | -0.61 | 0.5427 |

BLUPs for the Dental Data

Type 3 Tests of Fixed Effects

| Effect | Num DF | Den DF | F Value | Pr > F |
|-------------|-----------|-----------|---------|--------|
| GENDER | 2 | 54 | 384.72 | <.0001 |
| time*GENDER | 2 | 54 | 62.66 | <.0001 |

S
t
d
E
r

G
E
N
t
e
r

E
n
t
e
r

N
e
t
w
o
r
k

O
D
w
i
i
a
r
r
e
r

b
E
I
i
m
s
l
e
e
D
h
e
r
e
F
a
r
r
d

s
R
D
d
e
t
e
d

A
L
U
R

l
o
p
s
e

p
w
p
s
e

D
h
e
r
e
F
a
r
r
d

1 0 1 1 8 21.0 0 20.1783 0.43711 54 0.05 19.3019 21.0546 0.82175

2 0 1 1 10 20.0 0 21.0009 0.33796 54 0.05 20.3234 21.6785 -1.00095

3 0 1 1 12 21.5 0 21.8236 0.34908 54 0.05 21.1238 22.5235 -0.32365

4 0 1 1 14 23.0 0 22.6463 0.46259 54 0.05 21.7189 23.5738 0.35366

5 0 2 2 8 21.0 0 21.1527 0.43711 54 0.05 20.2763 22.0290 -0.15266

6 0 2 2 10 21.5 0 22.3957 0.33796 54 0.05 21.7181 23.0733 -0.89570

7 0 2 2 12 24.0 0 23.6387 0.34908 54 0.05 22.9389 24.3386 0.36126

8 0 2 2 14 25.5 0 24.8818 0.46259 54 0.05 23.9543 25.8092 0.61822

9 0 3 3 8 20.5 0 21.7737 0.43711 54 0.05 20.8974 22.6501 -1.27372

10 0 3 3 10 24.0 0 23.0873 0.33796 54 0.05 22.4098 23.7649 0.91266

11 0 3 3 12 24.5 0 24.4010 0.34908 54 0.05 23.7011 25.1008 0.09905

12 0 3 3 14 26.0 0 25.7146 0.46259 54 0.05 24.7871 26.6420 0.28543

...

...

...

...

BLUPs for the Dental Data

```

Model Information

Data Set                WORK.PROYUNIV
Dependent Variable      dist
Covariance Structures   Unstructured, Variance
                        Components
Subject Effects         newid, newid
Group Effect            GENDER
Estimation Method       ML
Residual Variance Method None
Fixed Effects SE Method Model-Based
Degrees of Freedom Method Containment
  
```

```

Class Level Information

Class    Levels    Values
newid      27      1 2 3 4 5 6 7 8 9 10 11 12 13
                        14 15 16 17 18 19 20 21 22 23
                        24 25 26 27
GENDER      2      0 1
  
```

```

Dimensions

Covariance Parameters      5
Columns in X               4
Columns in Z Per Subject   2
Subjects                   27
Max Obs Per Subject        4
  
```

BLUPs for the Dental Data

```

      Number of Observations
Number of Observations Read      108
Number of Observations Used      108
Number of Observations Not Used    0

      Iteration History
Iteration  Evaluations    -2 Log Like    Criterion

      0           1      478.24175986
      1           2      418.92503842      1.16632499
      2           1      416.18869903      1.23326209
      3           1      407.89638533      0.01954268
      4           2      406.88264563      0.00645800
      5           1      406.10632159      0.00056866
      6           1      406.04318997      0.00000764
      7           1      406.04238894      0.00000000

      Convergence criteria met.
      Covariance Parameter Estimates
Cov Parm    Subject    Group    Estimate

UN(1,1)     newid
UN(2,1)     newid
UN(2,2)     newid
Residual    newid      GENDER 0    0.4449
Residual    newid      GENDER 1    2.6294

      Fit Statistics

-2 Log Likelihood      406.0
AIC (smaller is better) 424.0
AICC (smaller is better) 425.9
BIC (smaller is better) 435.7

      Null Model Likelihood Ratio Test

      DF    Chi-Square    Pr > ChiSq

      4      72.20      <.0001

```

BLUPs for the Dental Data

Type 3 Tests of Fixed Effects

| Effect | Num DF | Den DF | F Value | Pr > F |
|-------------|-----------|-----------|---------|--------|
| GENDER | 2 | 54 | 384.72 | <.0001 |
| time*GENDER | 2 | 54 | 62.66 | <.0001 |

FIXED EFFECTS data

| Obs | Effect | GENDER | Estimate |
|-----|-------------|--------|----------|
| 1 | GENDER | 0 | 17.3727 |
| 2 | GENDER | 1 | 16.3406 |
| 3 | time*GENDER | 0 | 0.4795 |
| 4 | time*GENDER | 1 | 0.7844 |

RECONFIGURED FIXED EFFECTS DATA

| Obs | GENDER | fixint | fixslope |
|-----|--------|---------|----------|
| 1 | 0 | 17.3727 | 0.47955 |
| 2 | 1 | 16.3406 | 0.78438 |

BLUPs for the Dental Data

| Obs | Effect | newid | Estimate | gender |
|-----|-----------|-------|----------|--------|
| 1 | Intercept | 1 | -0.4853 | 0 |
| 2 | time | 1 | -0.06820 | 0 |
| 3 | Intercept | 2 | -1.1922 | 0 |
| 4 | time | 2 | 0.1420 | 0 |
| 5 | Intercept | 3 | -0.8535 | 0 |
| 6 | time | 3 | 0.1773 | 0 |
| 7 | Intercept | 4 | 1.7024 | 0 |
| 8 | time | 4 | 0.04017 | 0 |
| 9 | Intercept | 5 | 0.9136 | 0 |
| 10 | time | 5 | -0.08680 | 0 |
| 11 | Intercept | 6 | -0.6740 | 0 |
| 12 | time | 6 | -0.07292 | 0 |
| 13 | Intercept | 7 | -0.05461 | 0 |
| 14 | time | 7 | 0.03641 | 0 |
| 15 | Intercept | 8 | 1.9350 | 0 |
| 16 | time | 8 | -0.1149 | 0 |
| 17 | Intercept | 9 | -0.2190 | 0 |
| 18 | time | 9 | -0.1151 | 0 |
| 19 | Intercept | 10 | -2.9974 | 0 |
| 20 | time | 10 | -0.09085 | 0 |
| 21 | Intercept | 11 | 1.9249 | 0 |
| 22 | time | 11 | 0.1530 | 0 |
| 23 | Intercept | 12 | 1.3469 | 1 |
| 24 | time | 12 | 0.08788 | 1 |
| 25 | Intercept | 13 | -0.8676 | 1 |
| 26 | time | 13 | -0.04068 | 1 |
| 27 | Intercept | 14 | -0.3575 | 1 |
| 28 | time | 14 | -0.02176 | 1 |
| 29 | Intercept | 15 | 1.5946 | 1 |
| 30 | time | 15 | -0.02772 | 1 |
| 31 | Intercept | 16 | -1.1581 | 1 |
| 32 | time | 16 | -0.04153 | 1 |
| 33 | Intercept | 17 | 0.8972 | 1 |
| 34 | time | 17 | 0.02260 | 1 |
| 35 | Intercept | 18 | -0.6889 | 1 |
| 36 | time | 18 | -0.02853 | 1 |
| 37 | Intercept | 19 | -0.1443 | 1 |
| 38 | time | 19 | -0.07348 | 1 |
| 39 | Intercept | 20 | -0.1273 | 1 |
| 40 | time | 20 | 0.02544 | 1 |

BLUPs for the Dental Data

| | | | | |
|----|-----------|----|----------|---|
| 41 | Intercept | 21 | 2.5349 | 1 |
| 42 | time | 21 | 0.1088 | 1 |
| 43 | Intercept | 22 | -0.2261 | 1 |
| 44 | time | 22 | -0.08535 | 1 |
| 45 | Intercept | 23 | -0.6374 | 1 |
| 46 | time | 23 | 0.006510 | 1 |
| 47 | Intercept | 24 | -1.7008 | 1 |
| 48 | time | 24 | 0.1139 | 1 |
| 49 | Intercept | 25 | 0.2387 | 1 |
| 50 | time | 25 | -0.03166 | 1 |
| 51 | Intercept | 26 | 0.1180 | 1 |
| 52 | time | 26 | 0.06104 | 1 |
| 53 | Intercept | 27 | -0.8223 | 1 |
| 54 | time | 27 | -0.07545 | 1 |

| Obs | newid | gender | ranint | ranslope |
|-----|-------|--------|----------|----------|
| 1 | 1 | 0 | -0.48526 | -0.06820 |
| 2 | 2 | 0 | -1.19224 | 0.14198 |
| 3 | 3 | 0 | -0.85346 | 0.17726 |
| 4 | 4 | 0 | 1.70243 | 0.04017 |
| 5 | 5 | 0 | 0.91363 | -0.08680 |
| 6 | 6 | 0 | -0.67403 | -0.07292 |
| 7 | 7 | 0 | -0.05461 | 0.03641 |
| 8 | 8 | 0 | 1.93498 | -0.11486 |
| 9 | 9 | 0 | -0.21898 | -0.11515 |
| 10 | 10 | 0 | -2.99738 | -0.09085 |
| 11 | 11 | 0 | 1.92494 | 0.15297 |
| 12 | 12 | 1 | 1.34688 | 0.08788 |
| 13 | 13 | 1 | -0.86755 | -0.04068 |
| 14 | 14 | 1 | -0.35750 | -0.02176 |
| 15 | 15 | 1 | 1.59462 | -0.02772 |
| 16 | 16 | 1 | -1.15811 | -0.04153 |
| 17 | 17 | 1 | 0.89718 | 0.02260 |
| 18 | 18 | 1 | -0.68894 | -0.02853 |
| 19 | 19 | 1 | -0.14433 | -0.07348 |

BLUPs for the Dental Data

| | | | | |
|----|----|---|----------|----------|
| 20 | 20 | 1 | -0.12730 | 0.02544 |
| 21 | 21 | 1 | 2.53489 | 0.10877 |
| 22 | 22 | 1 | -0.22609 | -0.08535 |
| 23 | 23 | 1 | -0.63735 | 0.00651 |
| 24 | 24 | 1 | -1.70079 | 0.11392 |
| 25 | 25 | 1 | 0.23870 | -0.03166 |
| 26 | 26 | 1 | 0.11799 | 0.06104 |
| 27 | 27 | 1 | -0.82229 | -0.07545 |

| Obs | GENDER | fixint | fixslope | newid | ranint | ranslope | beta0i | betaii |
|-----|--------|---------|----------|-------|----------|----------|---------|---------|
| 1 | 0 | 17.3727 | 0.47955 | 1 | -0.48526 | -0.06820 | 16.8875 | 0.41135 |
| 2 | 0 | 17.3727 | 0.47955 | 2 | -1.19224 | 0.14198 | 16.1805 | 0.62152 |
| 3 | 0 | 17.3727 | 0.47955 | 3 | -0.85346 | 0.17726 | 16.5193 | 0.65681 |
| 4 | 0 | 17.3727 | 0.47955 | 4 | 1.70243 | 0.04017 | 19.0752 | 0.51971 |
| 5 | 0 | 17.3727 | 0.47955 | 5 | 0.91363 | -0.08680 | 18.2864 | 0.39274 |
| 6 | 0 | 17.3727 | 0.47955 | 6 | -0.67403 | -0.07292 | 16.6987 | 0.40662 |
| 7 | 0 | 17.3727 | 0.47955 | 7 | -0.05461 | 0.03641 | 17.3181 | 0.51595 |
| 8 | 0 | 17.3727 | 0.47955 | 8 | 1.93498 | -0.11486 | 19.3077 | 0.36469 |
| 9 | 0 | 17.3727 | 0.47955 | 9 | -0.21898 | -0.11515 | 17.1537 | 0.36440 |
| 10 | 0 | 17.3727 | 0.47955 | 10 | -2.99738 | -0.09085 | 14.3753 | 0.38869 |
| 11 | 0 | 17.3727 | 0.47955 | 11 | 1.92494 | 0.15297 | 19.2977 | 0.63251 |
| 12 | 1 | 16.3406 | 0.78438 | 12 | 1.34688 | 0.08788 | 17.6875 | 0.87225 |
| 13 | 1 | 16.3406 | 0.78438 | 13 | -0.86755 | -0.04068 | 15.4731 | 0.74369 |
| 14 | 1 | 16.3406 | 0.78438 | 14 | -0.35750 | -0.02176 | 15.9831 | 0.76262 |
| 15 | 1 | 16.3406 | 0.78438 | 15 | 1.59462 | -0.02772 | 17.9352 | 0.75665 |
| 16 | 1 | 16.3406 | 0.78438 | 16 | -1.15811 | -0.04153 | 15.1825 | 0.74285 |
| 17 | 1 | 16.3406 | 0.78438 | 17 | 0.89718 | 0.02260 | 17.2378 | 0.80697 |
| 18 | 1 | 16.3406 | 0.78438 | 18 | -0.68894 | -0.02853 | 15.6517 | 0.75584 |
| 19 | 1 | 16.3406 | 0.78438 | 19 | -0.14433 | -0.07348 | 16.1963 | 0.71090 |
| 20 | 1 | 16.3406 | 0.78438 | 20 | -0.12730 | 0.02544 | 16.2133 | 0.80981 |
| 21 | 1 | 16.3406 | 0.78438 | 21 | 2.53489 | 0.10877 | 18.8755 | 0.89315 |
| 22 | 1 | 16.3406 | 0.78438 | 22 | -0.22609 | -0.08535 | 16.1145 | 0.69903 |
| 23 | 1 | 16.3406 | 0.78438 | 23 | -0.63735 | 0.00651 | 15.7033 | 0.79088 |
| 24 | 1 | 16.3406 | 0.78438 | 24 | -1.70079 | 0.11392 | 14.6398 | 0.89830 |
| 25 | 1 | 16.3406 | 0.78438 | 25 | 0.23870 | -0.03166 | 16.5793 | 0.75272 |
| 26 | 1 | 16.3406 | 0.78438 | 26 | 0.11799 | 0.06104 | 16.4586 | 0.84542 |
| 27 | 1 | 16.3406 | 0.78438 | 27 | -0.82229 | -0.07545 | 15.5183 | 0.70893 |